

Thread Programmierung unter Windows mit Delphi

von Michael Puff

aus dem Inhalt:

- Grundlagen der Thread Programmierung (BeginThread, ...)
- Thread-Ablaufsteuerung (ResumeThread, SuspendThread)
- Thread-Prioritäten (SetPriorityClass, SetThreadPriority, ...)
- Thread-Synchronisation (CriticalSection, InterLockedExchangeXXX, ...)
- Thread-Synchronisation mit Kernelobjekten (neu in 2.2) (WaitForXXX, CreateEvent, SetEvent)
- Das Thread Objekt der VCL

Inhaltsverzeichnis

1	Vorwort.....	4
2	Einführung.....	5
	2.1 Geschichtliches.....	5
	2.2 Begriffsdefinition.....	5
	2.3 Wann sollte man Threads einsetzen und wann nicht?.....	7
3	Grundlagen.....	9
	3.1 Zur Veranschaulichung.....	9
	3.2 Die Thread-Funktion.....	10
	3.3 Einen Thread abspalten.....	10
	3.4 Beenden eines Threads.....	11
4	Thread-Ablaufsteuerung.....	15
	4.1 Anhalten und Fortsetzen von Threads.....	16
	4.2 Temporärer Wechsel zu einem anderen Thread.....	17
	4.3 Thread Ausführungszeiten, Beispielanwendung ThreadTimes.....	18
5	Thread-Prioritäten.....	20
	5.1 Darstellung der Prioritäten.....	20
	5.2 Programmieren von Prioritäten.....	24
	5.3 Beispielanwendung "Priority-Demo".....	27
6	Thread-Synchronisation.....	29
	6.1 Atomarer Variablenzugriff.....	29
	6.2 Kritische Abschnitte.....	31
7	Thread-Synchronisation mit Kernel-Objekten.....	36
	7.1 WaitForxxx-Funktionen.....	36
	7.2 Ereignisobjekte.....	39
8	Das VCL Thread-Objekt.....	41
	8.1 Erzeugen, Eigenschaften, Methoden.....	41
	8.2 Beispiel einer Execute Methode.....	43
	8.3 Synchronisation des Thread-Objektes.....	44

Kontakt:

Homepage : www.luckie-online.de

E-Mail : tutorials@luckie-online.de

Bezugsquellen / Downloadmöglichkeiten:

Dieses PDF Dokument und Sourcen der Demos: www.luckie-online.de

Quellenangabe:

- Richter, Jeffrey: Microsoft Windows Programmierung für Experten, 4. vollständig überarbeitete Ausgabe. Microsoft Press 2000
- Ruediger R. Asche: Multithreading für Rookies, Microsoft Developer Network Technologie Group, September 1993
- Martin Harvey: Multithreading - The Delphi Way, 1.1a, <http://www.pergolesi.demon.co.uk>

1 Vorwort

Um was soll es in diesem Tutorial gehen? Wie der Titel schon sagt um Threads und ihre Programmierung unter Windows mit Delphi. Dabei werde ich erst mal auf die Theorie eingehen und diese an Hand von Win32API Code veranschaulichen. Dabei werde ich folgende Themen ansprechen:

- Grundlagen (einen Thread erzeugen / abspalten. Einen Thread beenden.)
- Steuerung von Threads (Anhalten / Fortsetzen)
- Thread Prioritäten (Prozess-Prioritäten, Thread-Prioritäten. Programmieren von Prioritäten)
- Synchronisation von Threads (atomarer Variablenzugriff. Kritische Abschnitte)

Am Ende werde ich dann noch mal etwas detaillierter auf das Thread-Objekt der VCL eingehen.

Dem Tutorial liegen Demo-Programme zu allen Themengebieten bei, welche hier angesprochen werden. Der Code dieser Demo-Anwendungen wird sich in Auszügen auch hier im Dokument wiederfinden.

An dieser Stelle möchte ich mich auch noch mal recht herzlich bei Peter Rehders für das Korrekturlesen des Tutorials bedanken.

2 Einführung

2.1 Geschichtliches

Am Anfang, als die Threads laufen lernten, gab es kein Multitasking. Ja, man konnte es noch nicht mal Singletasking nennen. Man hat sein Programm in Lochkarten gestanzt, hat sie im Rechenzentrum abgegeben, und Tage später bekam man einen Stapel Lochkarten wieder zurück, mit oder auch oft ohne die gewünschten Resultate. War ein Fehler drin, musste die entsprechende Lochkarte ganz neu gestanzt werden.

Aber die Dinge haben sich weiterentwickelt. Das erste Konzept, bei dem mehrere Threads parallel ausgeführt wurden, tauchte bei den so genannten "time sharing systems" auf. Es gab einen großen zentralen Computer der mehrere Client Computer mit Rechenleistung versorgte. Man spricht von Mainframes und Workstations. Hier war es wichtig, dass die Rechenleistung bzw. die CPU-Zeit gerecht zwischen den Workstations aufgeteilt wurde. In diesem Zusammenhang erschien auch zum ersten Mal das Konzept von Prozessen und Threads. Desktop-Computer haben eine ähnliche Entwicklung durchgemacht. Frühe DOS- und Windows-Rechner waren Singletasking-Systeme. Das heißt, ein Programm lief exklusiv und allein für sich auf einem Rechner. Selbst das Betriebssystem bekam während der Ausführung eines Programms keine Rechenzeit zugeteilt. Aber die Anforderungen stiegen und die Rechner stellten immer mehr Leistung zur Verfügung. Leistung die auch heute noch die meiste Zeit, die der Rechner in Betrieb ist, brachliegt. Warum sie also nicht so effizient wie möglich nutzen, und den Rechner mehrere Dinge gleichzeitig machen lassen, wenn er kann? Das Multitasking war geboren.

2.2 Begriffsdefinition

2.2.1 Was ist ein Prozess?

Ganz allgemein kann man sagen, dass ein Prozess die laufende Instanz einer Anwendung, oder wenn man so will, eines Programms, ist.

Ein Prozess besteht immer aus zwei Teilen:

1. Dem Kernel-Objekt¹, welches dem Betriebssystem zur Verwaltung des Prozesses dient, und dem
2. Adressraum, der den ausführbaren Code, die Daten und die dynamischen Speicherzuweisungen (Stack- und Heap-Zuweisungen) enthält.

Jeder Prozess muss mindestens einen Thread besitzen, den primären Thread. Diesen primären Thread braucht man nicht selber zu erzeugen, dies übernimmt der Loader, der den Prozess startet.. Bei Windows Programmen, die ein Fenster besitzen, enthält der primäre Thread die Nachrichtenschleife und die Fenster-Prozedur der Anwendung. Tut er dies nicht, so gäbe es keine Existenzberechtigung für ihn und das Betriebssystem würde ihn mitsamt seines Adressraumes automatisch löschen. Dies wiederum bedeutet, wird der primäre Thread eines Pro-

¹ Ein Kernel-Objekt ist im Grunde genommen nichts weiter als ein Speicherblock, den der Kernel belegt hat. Dieser Speicherblock stellt eine Datenstruktur da, deren Elemente Informationen über das Objekt verwalten. Hinweis: Kernel-Objekte gehören dem Kernel, nicht dem Prozess.

zesses beendet, wird auch der zugehörige Prozess beendet, das Kernel-Objekt zerstört und der reservierte Adressraum wieder freigegeben. Ein Prozess dient also quasi als Container für den primären Thread und weiteren Threads, die im Laufe der des Programms abgespalten werden können.

Ein Prozess, oder genauer, dessen Threads führen den Code in einer geordneten Folge aus. Dabei operieren sie alle streng getrennt voneinander. Das heißt, ein Prozess kann nicht auf den Adressraum eines anderen Prozesses zugreifen. Ein Prozess sieht also die anderen Threads gar nicht und hat den Eindruck, als ob er alle Systemressourcen (Speicher, Speichermedien, Ein- / Ausgabe, CPU) für sich alleine hätte. In Wirklichkeit ist dies natürlich nicht der Fall, so dass das Betriebssystem die Aufgabe übernehmen muss, dem Prozess, dem seine eigenen "virtuellen" Ressourcen zur Verfügung stehen, reale Ressourcen zuzuordnen. In diesem Zusammenhang spricht man auch beispielsweise von virtuellen Prozessoren im System. Es gibt so viele virtuelle Prozessoren, wie es Prozesse / Threads im System gibt.

Auf der anderen Seite gibt es aber auch Ressourcen, die von den Prozessen geteilt werden können. Da wären zum Beispiel dynamische Bibliotheken, die DLL's, zu nennen. Eine DLL kann von mehreren Prozessen gleichzeitig genutzt werden. Dabei wird sie nur einmal geladen, ihr Code aber in den Adressraum des Prozesses eingeblendet, welcher die DLL nutzt. Dies kann man unter anderem zur Inter Process Communication nutzen. Dies hat aber des Weiteren noch den Vorteil, dass gleicher Code nur einmal geschrieben werden muss und gleichzeitig von mehreren Prozessen genutzt werden kann.

Das Besondere ist, dass nur allein der Kernel Zugriff auf diesen Speicherblock hat. Anwendungen können ihn weder lokalisieren, noch auf ihn zugreifen. Der Zugriff kann nur über API-Funktionen erfolgen. Dies gewährleistet die Konsistenz des Kernel-Objektes.

2.2.2 Threads, die arbeitende Schicht

Ein Thread beschreibt nun einen "Ausführungspfad" innerhalb eines Prozesses. Und der Thread ist es, der den eigentlichen Programmcode ausführt. Der Prozess dient, wie schon gesagt, lediglich als Container, der die Umgebung (den Adressraum) des Threads bereitstellt.

Threads werden immer im Kontext des ihnen übergeordneten Prozesses erzeugt. Das heißt, ein Thread führt seinen Code immer im Adressraum seines Prozesses aus. Werden beispielsweise zwei oder mehr Threads im Kontext eines Prozesses ausgeführt, teilen sie sich einen einzigen Adressraum, dem des Prozesses nämlich. Sie können also denselben Code ausführen und dieselben Daten bearbeiten. Den Zugriff auf den gleichen Adressraum kann man einerseits als Segen, andererseits auch als Fluch sehen. Denn es ist zwar einfach Daten zwischen den Threads auszutauschen, nur die Synchronisation des Datenaustausches kann mitunter recht schwierig werden.

Wann immer es möglich, wenn Multithreading gefragt ist, sollte man einen eigenen Thread starten, anstatt eines eigenen Prozesses, wie man es unter 16-Bit Windows noch tun musste. Dies kannte zwar Multitasking, aber kein Multithreading. Denn im Gegensatz zum Anlegen eines Prozesses, erfordert das Starten eines Threads weniger Systemressourcen, da nur das Thread-Kernel-Objekt angelegt und verwaltet werden muss. Das Anlegen und Verwalten eines Adressraumes fällt ja weg, da der Thread ja den Adressraum des übergeordneten Prozesses mitbenutzt. Einzig und allein ein weiterer Stack muss für den Thread im Adressraum des Prozesses angelegt werden.

2.2.3 Multitasking und Zeitscheiben

Wie setzt das Betriebssystem nun die Technik des Multitasking / Multithreading um? Das Problem ist ja, dass in heutigen herkömmlichen Desktop Computern meist nur ein Prozessor zur Verfügung steht. Diesen Prozessor müssen sich also nun die im System laufenden Prozesse teilen. Dafür gibt es zwei Ansätze unter Windows:

1. Kooperatives Multitasking (16-Bit Windows)
2. Präemptives Multitasking (32-Bit Windows)

Beim kooperativen Multitasking ist jeder Prozess selbst dafür verantwortlich Rechenzeit abzugeben und so anderen Prozessen die Möglichkeit zu geben weiter zu arbeiten. Die Prozesse müssen also "kooperieren". Wer hingegen beim präemptiven Multitasking die Zuteilung von Rechenzeit ganz alleine dem Betriebssystem obliegt. Anders wäre es beispielsweise nicht möglich gewesen den Taskmanager in das System einzufügen, denn wie will man einen anderen Prozess starten, wenn ein anderer den Prozessor in Beschlag nimmt und ihn nicht wieder freigibt? Da zudem der Taskmanager mit einer höheren Priorität ausgeführt wird, als andere Prozesse, kann er immer noch dazu genutzt werden "hängengebliebene" Prozesse zu beenden.

Beim präemptiven Multitasking wird die Rechenzeit, die zur Verfügung steht, in so genannte Zeitscheiben eingeteilt. Das heißt, ein Prozess kann die CPU für eine bestimmte Zeit nutzen, bevor das Betriebssystem die CPU diesem Prozess entzieht und sie einem anderen Prozess zuteilt. Dies gilt genauso für Threads innerhalb eines Prozesses. Denn unter 32-Bit-Windows ist die kleinste ausführbare Einheit ein Thread und nicht wie unter 16-Bit-Windows die Instanz eines Prozesses. Wird nun ein Thread unterbrochen, speichert das Betriebssystem den momentanen Zustand des Threads, sprich die Werte der CPU Register usw. und wenn dem Thread Rechenzeit zugeteilt wird, wird der Zustand der Register vom Betriebssystem wiederhergestellt und der Thread kann an der Stelle, an der er unterbrochen wurde, weiter Code ausführen. Tatsächlich wird also nichts parallel ausgeführt, sondern nacheinander. Nur erscheint die Ausführung für den Benutzer parallel, da die Zeitscheiben so kurz sind, dass es nicht auffällt und der Anschein von Parallelität erzeugt wird.

2.3 Wann sollte man Threads einsetzen und wann nicht?

Um es vorweg zu nehmen, es gibt keine festen Regeln, wann man mehrere Threads zu verwenden hat und wann nicht. Dies hängt ganz davon ab, was das Programm tun soll und wie wir das erreichen wollen. Man kann eigentlich nur vage Empfehlungen aussprechen. Was man allerdings jedem raten kann, der Multithread-Anwendungen entwickelt, ist, dass Threads umsichtig eingesetzt werden sollten. Denn: Die Implementation ist eigentlich trivial. Das eigentlich Schwere an der Sache ist, Threads interagieren zu lassen, ohne dass sie sich ins Gehege kommen.

Der Einsatz von Threads ist dann sinnvoll, wenn Code im Hintergrund ausgeführt werden soll, also Code, der keine Benutzereingaben erfordert und auch keine Ausgaben hat, die den Benutzer zum Zeitpunkt der Ausführung interessieren. Als Beispiele wären da zu nennen: Jegliche langen Berechnungen von irgendetwas, eine Folge von Primzahlen oder ähnlichem. Oder die automatische Rechtschreibkorrektur in Textverarbeitungen. Oder die Übertragung / das Empfangen von Daten über ein Netzwerk oder Port. Auch Code der asynchron ausgeführt wird, wie das Pollen eines Ports, ist besser in einem eigenen Thread aufgehoben, anstatt mit dem Vordergrund-Task zu kämpfen, der im gleichen Thread ausgeführt wird.

Ein Beispiel, wann Threads nicht sinnvoll sind: Nehmen wir an, wir haben eine Textverarbeitung und wir wollen eine Funktion zum Drucken implementieren. Schön wäre es, wenn selbige nicht die ganze Anwendung blockieren würde und man mit ihr weiter arbeiten könnte. Diese Situation scheint doch gerade zu prädestiniert für einen Thread zu sein, der das Dokument ausdruckt. Nur bei etwas genaueren Überlegungen stößt man auf Probleme. Wie druckt man ein Dokument aus, welches sich ständig ändert, weil daran weitergearbeitet wird? Eine Lösung wäre, das Dokument für die Dauer des Drucks zu sperren und dem Benutzer nur an Dokumenten arbeiten zu lassen, die sich nicht im Druck befinden. Unschön. Aber wie sieht es mit der Lösung aus? Man speichert das zu druckende Dokument in einer temporären Datei und druckt diese? Ein Thread wäre dazu nicht mehr nötig.

3 Grundlagen

3.1 Zur Veranschaulichung

Veranschaulichen wir uns das ganze mal an Hand einer Grafik.




Abbildung 1 Veranschaulichung von Threads

Legende:

- schräge Striche kennzeichnen den Beginn eines Threads
- durchgezogene Linien, wenn der Thread Code ausführt
- gestrichelte Linien, wenn sich der Thread im Wartezustand befindet
- Punkte bezeichnen das Ende eines Threads

Zuerst wird der Hauptthread gestartet. Dies übernimmt der Loader, wenn der Prozess erzeugt wird. Als nächstes befindet sich der Hauptthread im Wartezustand, er führt also keinen Code aus. Der Benutzer tätigt keine Eingaben und macht auch sonst nichts mit dem Fenster. Ein Klick auf eine Schaltfläche weckt den Hauptthread und veranlasst ihn Code auszuführen. In diesem Fall wird ein zweiter Thread abgespalten. Dieser wird aber nicht gleich gestartet, sondern verbringt auch erst eine Zeit im Wartezustand bevor er von Hauptthread gestartet wird.

Der folgende Abschnitt der Grafik verdeutlicht wie Code von beiden Prozessen parallel ausgeführt wird. Letztendlich enden beide Threads, wobei der zweite Thread vorher entweder

automatisch endet, weil er seinen Code abgearbeitet hat oder sonst wie (dazu später) beendet wurde.

3.2 Die Thread-Funktion

Jeder Thread braucht eine Funktion, die den Code enthält, welchen der Thread ausführen soll. Diese bezeichnet man auch als Startfunktion des Threads. Die Delphi Deklaration sieht wie folgt aus:

```
type TThreadFunc = function (Parameter: Pointer): Integer;
```

Als Parameter kann optional ein Pointer übergeben werden, der auf eine Datenstruktur zeigt mit Werten, die dem Thread zum Arbeiten übergeben werden sollen. Der Rückgabewert der Thread-Funktion ist eine Ganzzahl, welche beim Beenden dem Exitcode des Threads entspricht. Im Demo Programm "ThreadTimes" wird demonstriert, wie man das macht.

Zu beachten ist allerdings noch folgendes: Benutzt man `BeginThread` als Wrapper für die API-Funktion `CreateThread`, so darf man nicht die Aufrufkonvention `stdcall` für die Thread-Funktion benutzen. Grund: `BeginThread` benutzt eine "Thread-Wrapper"-Funktion anstatt des wirklichen Einsprungpunktes. Diese Funktion ist als `stdcall` definiert, sie verschiebt die Parameter auf dem Stack und ruft dann die eigene Thread-Funktion auf.

Des Weiteren sollte man möglichst keine ungeschützten Zugriffe auf globale Variablen durchführen, da ein simultaner Zugriff mehrerer Threads den Variablen-Inhalt "beschädigen" könnte. Parameter und lokale Variablen hingegen werden auf dem thread-eigenen Stack abgelegt und sind dort weniger anfällig für "Beschädigungen". Wie man trotzdem einen sicheren Zugriff auf globale Variablen realisiert wird im Kapitel sechs besprochen.

3.3 Einen Thread abspalten

Wie schon gesagt, wird der primäre Thread vom Loader automatisch angelegt. Will man nun in seinem Prozess einen weiteren Thread erzeugen, so muss man auf die Funktion `CreateThread` zurückgreifen:

```
function CreateThread(lpThreadAttributes: Pointer; dwStackSize: DWORD;  
  lpStartAddress: TFNThreadStartRoutine; lpParameter: Pointer;  
  dwCreationFlags: DWORD; var lpThreadId: DWORD): THandle; stdcall;
```

In der folgenden Tabelle findet sich eine kurze Erklärung der Parameter.

Parameter	Bedeutung
lpThreadAttributes	Zeiger auf eine Struktur, welche festlegt, ob das zurückgegebene Handle vererbt werden kann oder nicht.
dwStackSize	Legt die Größe des Stacks fest. 0 überlässt die Festlegung Windows.
lpStartAddress	Zeiger auf die Thread-Funktion.
lpParameter	Zeiger auf eine Variable, die dem Thread beim Start übergeben werden soll.
dwCreationFlags	Ein Flag, welches festlegt wie der Thread erzeugt werden soll. Dieser Flag kann entweder CREATE_SUSPENDED sein, der Thread wird im angehaltenen Zustand erzeugt oder 0, dann startet der Thread sofort nach seiner Erzeugung.
var lpThreadId	Eindeutiger Bezeichner des Threads.

Tabelle 1 Parameter von CreateThread

Was passiert nun nach dem Aufruf von CreateThread? Das System legt ein Kernel-Objekt für den Thread an, um ihn verwalten zu können. Des Weiteren wird im Adressraum des übergeordneten Prozesses ein separater Stack für den Thread angelegt.

Der neue Thread wird im selben Kontext ausgeführt wie der Thread, der ihn erzeugte. Folge: Der neue Thread kann auf alle prozess-eigenen Handles für Kernel-Objekte, den gesamten prozess-eigenen Speicher, sowie auf alle Stacks aller anderen Threads innerhalb des Prozesses zugreifen.

In einem Delphi-Programm sollten Sie nie die Funktion CreateThread direkt aufrufen. Benutzen Sie stattdessen die Funktion BeginThread. Grund: BeginThread kapselt zwar nur die API-Funktion CreateThread, setzt aber zusätzlich noch die globale Variable IsMultiThread und macht somit den Heap thread-sicher.

3.4 Beenden eines Threads

Ein Thread kann auf vier Arten beendet werden:

1. Die Thread-Funktion endet.
2. Der Thread beendet sich selbst mit der Funktion ExitThread
3. Der Thread wird durch TerminateThread beendet.
4. Der Prozess des Threads endet.

3.4.1 Die Thread-Funktion endet

Threads sollten so entworfen werden, dass sie nach Ausführung ihres Codes selbstständig enden. Grundsätzlich ist es so, dass der Thread automatisch endet, wenn er seinen Code abgearbeitet hat. Handelt es sich um einen Thread, der die ganze Zeit etwas machen soll, also einen Port pollen oder so, dann wird dieser Code normalerweise in eine Schleife gesteckt. Diese Schleife kann man mit einem break oder sonst einer Abbruchbedingung verlassen und so

den Thread sich selber enden lassen:

```
while bRunning <> 0 do
begin
  ...;
  ...;
end;
```

Nutzt man das Thread-Objekt der VCL kann man alternativ die Eigenschaft Terminated abfragen, welche auch von außerhalb des Threads gesetzt werden kann:

```
for Loop := 0 to ... do
begin
  ...;
  if Terminated then
    break;
  ...;
end;
```

Ein Thread sollte sich deshalb selber beenden, weil dann sichergestellt ist, dass

- der thread-eigene Stack wieder freigegeben wird
- der ExitCode den Rückgabewert der Thread-Funktion enthält
- der Zugriffszähler des thread-bezogenen Kernel-Objektes decremementiert wird
- alle Referenzen auf geladene DLL's und ähnlichem decremementiert werden

3.4.2 ExitThread

Eine weitere Möglichkeit einen Thread sich selber beenden zu lassen, ist die Verwendung der API Funktion ExitThread.

```
procedure ExitThread(dwExitCode: DWORD); stdcall;
```

Parameter	Bedeutung
dwExitCode	Exitcode des Threads

Tabelle 2 Parameter von ExitThread

ExitThread hat keinen Rückgabewert, da nach Aufruf, der Thread keinen Code mehr ausführen kann.

Auch hier werden alle Ressourcen wieder freigegeben.

Hinweis: In einer Delphi Anwendung sollte statt ExitThread das Äquivalent zu BeginThread EndThread aufgerufen werden.

3.4.3 TerminateThread

Im Gegensatz zu ExitThread, welches nur den aufrufenden Thread beenden kann (schon zu

erkennen an der Tatsache, dass `ExitThread` kein Parameter übergeben werden kann, welcher den zu beendenden Thread näher identifiziert.), kann mit `TerminateThread` jeder Thread beendet werden, dessen Handle man hat.

```
function TerminateThread(hThread: THandle; dwExitCode: DWORD): BOOL; stdcall;
```

Parameter	Bedeutung
<code>hThread</code>	Handle des Threads der beendet werden soll.
<code>dwExitCode</code>	Exitcode des Threads.

Tabelle 3 Parameter von `TerminateThread`

`TerminateThread` arbeitet asynchron. Das heißt, `TerminateThread` kehrt sofort zurück. Es teilt dem Betriebssystem nur mit, dass der betreffende Thread beendet werden soll. Wann letztendlich das Betriebssystem den Thread beendet, ob überhaupt (fehlende Rechte könnten dies verhindern), steht nicht fest. Kurz: Nach Rückkehr der Funktion ist nicht garantiert, dass der betreffende Thread auch beendet wurde.

Des Weiteren ist zu beachten, dass alle vom Thread belegten Ressourcen erst dann wieder freigegeben werden, wenn der übergeordnete Prozess endet. Sprich der thread-eigene Stack wird zum Beispiel nicht abgebaut.

Abschließend kann man sagen, dass man vermeiden sollte einen Thread gewaltsam zu beenden. Und dies aus zwei Gründen:

1. Die belegten Ressourcen werden nicht freigegeben
2. Man weiß nicht, in welchem Zustand sich der Thread gerade befindet. Hat er beispielsweise gerade eine Datei exklusiv geöffnet, können auf diese Datei so lange keine anderen Prozesse zugreifen bis der Prozess des "abgeschossenen" Threads beendet wurde.

3.4.4 Vorgänge beim Beenden eines Threads

Wird ein Thread "sauber" beendet,

- werden alle zum Thread gehörenden Handles für User-Objekte freigegeben,
- der Exitcode wechselt von `STIIL_ACTIVE` auf den Rückgabewert der Thread-Funktion oder dem von `ExitThread` oder `TerminateThread` festgelegten Wert,
- wird der übergeordnete Prozess beendet, sofern es sich um den letzten Thread des Prozesses handelt und
- der Zugriffszähler des Thread-Objektes wird um eins verringert.

Mit der Funktion `GetExitCodeThread` lässt sich der Exitcode eines Threads ermitteln.

```
function GetExitCodeThread(hThread: THandle; var lpExitCode: DWORD): BOOL; stdcall;
```

Parameter	Bedeutung
hThread	Handle des Threads, dessen Exitcode ermittelt werden soll.
var lpExitCode	Out Parameter der den Exitcode nach Aufruf der Funktion enthält.

Tabelle 4 Parameter von GetExitCodeThread

So lange der Thread beim Aufruf noch nicht terminiert ist, liefert die Funktion den Wert STILL_ACTIVE (\$103) zurück, ansonsten den Exitcode des betreffenden Threads.

4 Thread-Ablaufsteuerung

Beim präemptiven Multitasking – im Gegensatz zum kooperativen Multitasking, wie es bei Windows der Fall ist, wird der Zeitraum in Scheiben definierter Länge unterteilt. Den einzelnen Prozessen werden durch den Scheduler Zeitscheiben zugewiesen. Läuft ihre Zeitscheibe ab, so werden sie unterbrochen und ein anderer Thread erhält eine Zeitscheibe.

Daher ist eine wesentliche Voraussetzung für echtes präemptives Multitasking eine CPU, die verschiedene Berechtigungs-Modi beherrscht und ein Betriebssystemkern, der in der Lage ist, die verschiedenen Modi dahingehend zu nutzen, Anwendungssoftware mit unterschiedlichen Berechtigungen auszuführen. Ein einfaches Beispiel:

- Anwendungssoftware ist "unterprivilegiert" im Sinne von "sie darf nicht alles". Sie muss nämlich von einem Prozess mit höher Berechtigungsstufe unterbrechbar sein. Das ist die Grundvoraussetzung dafür, dass der Kernel - wie oben beschrieben - ihr
 1. eine Zeitscheibe zuteilen und den Prozess aufrufen kann,
 2. bei Ablauf der Zeitscheibe den Prozess einfrieren und "schlafen" legen kann und
 3. bei Zuteilung einer neuen Zeitscheibe den Prozess wieder so aufwecken kann, dass der Prozess nicht merkt, dass er unterbrochen wurde
- Der Systemkern selbst läuft in einem Modus mit maximaler Berechtigungsstufe.
 1. 1.Bei Linux unterscheidet man hier zwischen Kernel-Mode und User-Mode,
 2. 2.bei Windows NT und Nachfolgern spricht man von "Ring 0" als Kernel-Modus und Ringen höherer Nummerierung für "unterprivilegierte" Prozesse.

Die Kontextstruktur des Threads enthält den Zustand der Prozessorregister zum Zeitpunkt der letzten Unterbrechung der Thread Ausführung. Das System überprüft nun ca. alle 20 Millisekunden alle vorhandenen Thread-Objekte, dabei werden nur einige Objekte als zuteilungsfähig und damit rechenbereit befunden. Es wird nun ein zuteilungsfähiges Thread-Objekt gewählt und der Inhalt der Prozessorregister aus der Kontextstruktur des Threads in die Register der CPU geladen. Diesen Vorgang bezeichnet man als Kontextwechsel.

Im aktiven Zustand nun, führt ein Thread Code aus und bearbeitet Daten. Nach 20 Millisekunden werden die Registerinhalte wieder gespeichert, der Thread befindet sich nun nicht mehr im aktiven Zustand der Ausführung und ein neues Thread-Objekt wird gewählt. Bei der Rechenzeit zuteilung werden nur zuteilungsfähige Threads berücksichtigt, das heißt Threads mit einem Unterbrechungszähler gleich 0. Besitzt ein Thread einen Unterbrechungszähler größer 0, bedeutet dies, dass er angehalten wurde und das ihm keine Prozessorzeit zugeteilt werden kann. Des weiteren bekommen auch Threads ohne Arbeit keine Rechenzeit, also Threads, die zum Beispiel auf eine Eingabe warten. Startet man zum Beispiel den Editor und gibt nichts ein, hat der Editor-Thread auch nichts zu tun und braucht auch keine Rechenzeit. Sobald aber etwas mit dem Editor geschieht, man nimmt eine Eingabe vor, oder man verschiebt das Fenster oder sollte das Fenster seine Neuzeichnung veranlassen müssen, so wird der Editor-Thread automatisch in den zuteilungsfähigen Zustand überführt. Das heißt nun aber nicht, dass er sofort Rechenzeit zugeteilt bekommt. Es heißt nur, dass er jetzt arbeitswillig ist und das System ihm bald Rechenzeit gewährt.

4.1 Anhalten und Fortsetzen von Threads

4.1.1 Fortsetzen

Ein Element im thread-bezogenen Kernel-Objekt stellt der Unterbrechungszähler dar. Beim Erzeugen eines Threads wird er mit 1 initialisiert, der Thread ist somit nicht zuteilungsfähig. Dies stellt sicher, dass er erst ausführungsbereit ist, wenn er vollständig initialisiert ist.

Nach der Initialisierung wird der Flag `CREATE_SUSPENDED` (5. Parameter von `CreateThread`) überprüft. Ist er nicht gesetzt, wird der Unterbrechungszähler decrementiert und somit auf 0 gesetzt, der Thread ist zuteilungsfähig. Erzeugt man einen Thread im angehaltenen Zustand, so hat man die Möglichkeit, seine Umgebung, zum Beispiel die Priorität, zu ändern. Er kann dann mit `ResumeThread` in den zuteilungsfähigen Zustand versetzt werden.

```
function ResumeThread(hThread: THandle): DWORD; stdcall;
```

Parameter	Bedeutung
hThread	Handle des Threads

Tabelle 5 Parameter von `ResumeThread`

`ResumeThread` gibt bei erfolgreicher Ausführung den bisherigen Wert des Unterbrechungszählers zurück oder `$FFFFFFFF` im Fehlerfall.

Das Anhalten eines Threads ist akkumulativ, das heißt, wird ein Thread dreimal angehalten, muss auch dreimal `ResumeThread` aufgerufen werden, um ihn wieder in den zuteilungsfähigen Zustand zu versetzen.

4.1.2 Anhalten

Das Gegenstück zu `ResumeThread` ist `SuspendThread`. Mit dieser Funktion kann ein Thread angehalten werden.

```
function SuspendThread(hThread: THandle): DWORD; stdcall;
```

Parameter	Bedeutung
hThread	Handle des Threads

Tabelle 6 Parameter von `SuspendThread`

`SuspendThread` gibt, wie auch `ResumeThread`, im Erfolgsfall den bisherigen Wert des Unterbrechungszählers zurück. Beide Funktionen sind von jedem Thread aus aufrufbar, sofern man über das Handle des betreffenden Threads verfügt.

Analog lauten die Methoden des Thread-Objektes `Resume` und `Suspend`.

Hinweis: Ein Thread kann sich mit `SuspendThread` auch selber anhalten. Er ist aber nicht in der Lage seine Fortsetzung zu veranlassen.

Man sollte allerdings vorsichtig mit dem Aufruf von `SuspendThread` sein, da nicht bekannt ist was der Thread gerade tut. Reserviert er zum Beispiel gerade Speicher auf dem Heap und wird dabei unterbrochen, dann würde die Unterbrechung des Threads eine Sperrung des Heapzugriffes bewirken. Würden nun anschließend andere Threads versuchen auf den Heap zuzugreifen, würde ihre Ausführung so lange blockiert bis der andere Thread fortgesetzt wird und seinen Zugriff beenden kann.

4.1.3 Zeitlich begrenztes Unterbrechen

Ein Thread kann auch dem System mitteilen, dass er für eine bestimmte Zeitdauer nicht mehr zuteilungsfähig sein möchte. Dazu dient die Funktion `Sleep`.

```
procedure Sleep(dwMilliseconds: DWORD); stdcall;
```

Parameter	Bedeutung
dwMilliseconds	Dauer in Millisekunden der Unterbrechung.

Tabelle 7 Parameter von `Sleep`

Diese Prozedur veranlasst den Thread sich selbst, so lange anzuhalten wie in `dwMilliseconds` angegeben. Dabei gilt zu beachten: Wird `Sleep` mit 0 aufgerufen, verzichtet der Thread freiwillig auf den Rest seiner Zeitscheibe.

Die Zuteilungsfähigkeit wird nur um ungefähr die angegebene Dauer in Millisekunden verzögert. Da es immer darauf ankommt, was sonst noch im System los ist.

Ruft man `Sleep` mit dem Parameter `INFINITE` auf, wird Thread überhaupt nicht mehr zuteilungsfähig. Ist aber nicht sehr sinnvoll, da es besser wäre ihn zu beenden und somit die Ressourcen wieder freizugeben.

4.2 Temporärer Wechsel zu einem anderen Thread

Mit der Funktion `SwitchToThread` stellt das System eine Funktion bereit, um zu einem anderen Thread umzuschalten. Gibt es allerdings keinen zuteilungsfähigen Thread, kehrt `SwitchToThread` sofort zurück.

```
function SwitchToThread: BOOL; stdcall;
```

`SwitchToThread` entspricht fast dem Aufruf von `Sleep(0)` mit dem Unterschied, dass `SwitchToThread` die Ausführung von Threads mit niedriger Priorität gestattet, während `Sleep(0)` den aufrufenden Thread sofort neu aktiviert, auch wenn dabei Threads mit niedriger Priorität verdrängt werden. Hinweis: Windows98 besitzt keine sinnvolle Implementation von `SwitchToThread`.

4.3 Thread Ausführungszeiten, Beispielanwendung ThreadTimes

Es kann vorkommen, dass man ermitteln will wie viel Zeit ein Thread für die Durchführung einer bestimmten Aufgabe benötigt. Oft findet man zu diesem Zweck Code wie diesen:

```
var
  StartTime, TimeDiff: DWORD;
begin
  StartTime := GetTickCount();

  // zu untersuchender Code

  TimeDiff := GetTickCount() - Starttime;
```

Mit diesem Code geht man allerdings davon aus, dass der Thread nicht unterbrochen wird, was aber bei einem präemptiven Betriebssystem nicht der Fall sein wird. Wird einem Thread die CPU entzogen, ist es entsprechend schwierig zu ermitteln, wie lange der Thread für die Erledigung verschiedener Aufgaben benötigt. Windows stellt allerdings eine Funktion bereit, die den Gesamtbetrag an Rechenzeit zurück gibt, die dem Thread tatsächlich zugeteilt worden ist. Diese Funktion heißt: `GetThreadTimes`.

```
function GetThreadTimes(hThread: THandle; var lpCreationTime, lpExitTime,
  lpKernelTime, lpUserTime: TFileTime): BOOL; stdcall;
```

Die Funktion gibt vier verschiedene Zeitwerte zurück:

Parameter	Bedeutung
CreationTime ²	Zeitpunkt, an dem der Thread erzeugt wurde.
ExitTime ²	Zeitpunkt, an dem der Thread beendet wurde. Wird der Thread noch ausgeführt, ist der Wert undefiniert.
KernelTime ³	... gibt an wie viel Zeit der Thread Code des Betriebssystems ausgeführt hat.
UserTime ³	... gibt an wie viel Zeit der Thread Anwendungscode ausgeführt hat.

Tabelle 8 Parameter von `GetThreadTimes`

Mit Hilfe dieser Funktion lässt sich nun genau ermitteln wie viel Zeit der Thread wirklich mit dem Ausführen von Code verbracht hat. Auszug aus dem Demo `ThreadTimes`:

```
function Thread(p: Pointer): Integer;
var
  lblCounter: TStaticText;
  lblTickCount, lblThreadTimes: TStaticText;
  Counter: Integer;
  TickCountStart, TickCountEnd: Cardinal;
  KernelTimeStart, KernelTimeEnd: FileTime;
  UserTimeStart, UserTimeEnd: FileTime;
  Dummy: TFileTime;
  KernelTimeElapsed, UserTimeElapsed, TotalTime: Int64;
begin
  // Variablen initialisieren
  result := 0;
```

² Zeit seit dem 1. Januar 1601 in Nanosekunden-Intervallen

³ Wert in 100 Nanosekunden Intervallen

```

Counter := 1;
lblCounter := PThreadParameter(p)^.FLabel;
lblTickCount := PThreadParameter(p)^.FlblTickCount;
lblThreadTimes := PThreadParameter(p)^.FlblThreadTimes;
// Anfangszeit stoppen
TickCountStart := GetTickCount();
GetThreadTimes(GetCurrentThread, Dummy, Dummy, KernelTimeStart, UserTimeStart);
// Code ausführen
while (Counter < MAXCOUNT + 1) and (bRunning = 0) do
begin
  lblCounter.Caption := IntToStr(Counter);
  Inc(Counter);
  sleep(1);
end;
// Endzeit stoppen
GetThreadTimes(GetCurrentThread, Dummy, Dummy, KernelTimeEnd, UserTimeEnd);
TickCountEnd := GetTickCount();
// mit FileTime soll man laut PSDK nicht rechnen,
// Konvertierung nach Int64
// Berechnung der Differenz
KernelTimeElapsed := Int64(KernelTimeEnd) - Int64(KernelTimeStart);
UserTimeElapsed := Int64(UserTimeEnd) - Int64(UserTimeStart);
TotalTime := (KernelTimeElapsed + UserTimeElapsed) div 10000;
// Ausgabe der Ergebnisse
lblTickCount.Caption := IntToStr(TickCountEnd - TickCountStart);
lblThreadTimes.Caption := IntToStr(TotalTime);
// freigeben des Speichers für die Struktur auf der der zeiger zeigt
FreeMem(p);
end;

```

Die Beispielanwendung "ThreadTimes" demonstriert dies. In einem Label wird ein Zähler hochgezählt. Betätigt man die Schaltfläche "Stop", wird die Zählschleife verlassen und die Ausführungszeiten genommen. Ein Sleep(0) sorgt dafür, dass man auch bei relativ kurzen Zeitspannen etwas "sieht".




Abbildung 2 Beispielanwendung "ThreadTimes"

5 Thread-Prioritäten

Wie schon gesagt, bekommt jeder Thread ungefähr 20 Millisekunden Rechenzeit der CPU. Dies gilt allerdings nur, wenn alle Threads die gleiche Priorität hätten, dem ist aber nicht so. Jeder Thread besitzt auch eine Priorität, das heißt "Wichtigkeit". Diese Priorität bestimmt, welcher Thread als nächstes von der Ablaufsteuerung bei der Zuteilung von Rechenzeit berücksichtigt wird.

Windows kennt Prioritätsstufen zwischen 0 und 31, wobei 0 die niedrigste Priorität ist und 31 die höchste. Die Ablaufsteuerung berücksichtigt nun erst alle Threads mit der Prioritätsstufe 31. Findest es keine zuteilungsfähigen Threads mehr mit der Prioritätsstufe 31, kommen die anderen Threads dran. Jetzt könnte man davon ausgehen, dass Threads mit niedrigerer Priorität gar keine Rechenzeit mehr bekommen. Dies trifft nicht zu, da sich die meisten Threads im System im nichtzuteilungsfähigen Zustand befinden.

Beispiel:

So lange in der Nachrichtenschlange keine Nachricht auftaucht, die von GetMessage übermittelt werden könnte, verbleibt der primäre Thread im angehaltenen Zustand. Wird nun eine Nachricht eingereicht, merkt das System dies und versetzt den Thread in den zuteilungsfähigen Zustand. Wird nun kein zuteilungsfähiger Thread mit einer höheren Priorität gefunden, wird dem Thread Rechenzeit zugewiesen.

5.1 Darstellung der Prioritäten

Die Windows-API definiert eine abstrakte Schicht oberhalb der Ablaufsteuerung, das heißt die Anwendung / der Programmierer kommuniziert niemals direkt mit der Ablaufsteuerung. Stattdessen werden Windows-Funktionen aufgerufen, die die übergebenen Parameter je nach verwendeten Betriebssystem interpretieren.

Bei der Wahl der Prioritätsklasse sollte man unter anderem berücksichtigen, welche anderen Anwendungen eventuell noch parallel mit der eigenen ausgeführt werden. Des weiteren hängt die Wahl der Prioritätsklasse auch von der Reaktionsfähigkeit des Threads ab. Diese Angaben sind sehr vage, aber sie hängen eben von der individuellen Aufgabe des Threads ab und müssen dementsprechend gewählt werden.

5.1.1 Prozess-Prioritäten

Windows kennt sechs Prioritätsklassen: "Leerlauf", "Normal", "Niedriger als normal", "Normal", "Höher als normal", "Hoch", "Echtzeit".

Prioritäts-klasse	Beschreibung
Echtzeit	Die Threads in diesem Prozess müssen unmittelbar auf Ereignisse reagieren können, um zeitkritische Aufgaben durchführen zu können. Threads dieser Klasse, können sogar mit Threads des Betriebssystems konkurrieren. Nur mit äußerster Vorsicht benutzen.
Hoch	Es gilt das gleiche wie für die Klasse "Echtzeit", nur dass Prozesse mit dieser Priorität unter der Priorität von "Echtzeit" liegen. Der Taskmanger wird zum Beispiel mit dieser Priorität ausgeführt, um eventuell noch andere Prozesse ("entgleiste" Jeffrey Richter) beenden zu können.
Höher als normal	Die Ausführung erfolgt mit einer Priorität zwischen "Normal" und "Hoch". (neu in Windows2000)
Normal	Es werden keine besonderen Anforderungen an die Rechenzeitverteilung gestellt.
Niedriger als normal	Die Ausführung erfolgt mit einer Priorität zwischen "Normal" und "Leerlauf". (neu in Windows2000)
Leerlauf	Threads in diesem Prozess werden nur ausgeführt, wenn es sonst nichts zu tun gibt. Beispielsweise Bildschirmschoner oder Hintergrundprogramme wie der Indexdienst.

Tabelle 9 Beschreibung der Prozess-Prioritätsklassen

Die Prioritätsklasse "Hoch" sollte nur dann Verwendung finden, wenn es unbedingt nötig ist, da auch Systemprozesse mit dieser Prioritätsklasse laufen. Die Threads des Explorers werden mit dieser Priorität ausgeführt. Dadurch wird gewährleistet, dass die Shell eine hohe Reaktionsfähigkeit besitzt und noch entsprechend reagiert, wenn andere Prozesse ansonsten das System belasten. Da die Explorer-Threads aber die meiste Zeit im angehaltenen Zustand verweilen belasten sie das System nicht, sind aber trotzdem "zur Stelle" wenn der Benutzer sie braucht.

Die Prioritätsklasse "Echtzeit" sollte hingegen sogar vermieden werden, da sie dann eine höhere Priorität als die System-Threads / -Prozesse hätten und sie so das Betriebssystem bei der Durchführung seiner Aufgaben behindern würden. Es könnten zum Beispiel E/A Operationen oder Netzwerkübertragungen ausgebremst werden. Auch Tastatur- und / oder Mauseingaben könnten unter Umständen nicht schnell genug verarbeitet werden, so dass für den Benutzer der Eindruck entstünde, dass das System blockiert sei. Die Prioritätsklasse "Echtzeit" sollte nur dann Verwendung finden, wenn der Thread / Prozess die meiste Zeit nicht zuteilungsfähig ist und / oder der auszuführende Code schnell abgearbeitet ist.

Hinweis: Ein Prozess kann nur dann die Prioritätsklasse "Echtzeit" zugewiesen werden, wenn der angemeldete Benutzer über die Berechtigung "Anheben der Zeitplanungspriorität" verfügt. Standardmäßig haben diese Berechtigung Mitglieder der Gruppe Administratoren und Hauptbenutzer.

Die meisten Threads gehören der Prioritätsklasse "Normal" an. Die Prioritätsklassen "Höher als normal" und "Niedriger als normal" wurden von Microsoft neu bei Windows2000 hinzugefügt, weil sich Firmen beklagt hatten, dass die bisher verfügbaren Prioritäten nicht flexibel genug wären.

Hier mal eine Abbildung des Taskmanagers von Windows 2000 mit Darstellung der Basisprio-

ritätsklassen der Prozesse:

Name	PID	CPU-Nutzung (%)	CPU-Zeit	Speicher...	Basispriorität	Threads
Leerlaufprozess	0	99	1:12:27	16 KB	N/A	1
System	8	00	0:00:02	220 KB	Normal	34
SMSS.EXE	144	00	0:00:00	344 KB	Normal	6
CSRSS.EXE	168	00	0:00:07	2.368 KB	Hoch	9
WINLOGON.EXE	188	00	0:00:00	2.036 KB	Hoch	14
SERVICES.EXE	216	00	0:00:00	5.568 KB	Normal	37
LSASS.EXE	228	00	0:00:00	1.288 KB	Normal	14
svchost.exe	408	00	0:00:00	3.584 KB	Normal	10
spoolsv.exe	436	00	0:00:00	4.264 KB	Normal	12
AVWUPSRV.EXE	464	00	0:00:00	984 KB	Normal	3
svchost.exe	484	00	0:00:00	9.168 KB	Normal	26
kpf3.exe	500	00	0:00:00	5.092 KB	Normal	8
explorer.exe	504	00	0:00:01	3.892 KB	Normal	7
nsvsvc32.exe	560	00	0:00:00	1.044 KB	Normal	3
mstask.exe	580	00	0:00:00	3.108 KB	Normal	6
regsvc.exe	588	00	0:00:00	1.004 KB	Normal	2
WinMgmt.exe	668	00	0:00:03	616 KB	Normal	4
svchost.exe	680	00	0:00:00	6.280 KB	Normal	5
kpf3gui.exe	788	00	0:00:00	3.508 KB	Normal	3
explorer.exe	1052	00	0:00:01	4.236 KB	Normal	11
ICQLite.exe	1068	00	0:00:00	3.268 KB	Normal	7
TASKMGR.EXE	1112	00	0:00:00	2.104 KB	Hoch	3
soffice.exe	1136	00	0:01:10	41.908 KB	Normal	10

Prozesse: 23 CPU-Nutzung: 0% Speichernutzung: 112448 KB / 1278444 KB

Abbildung 3 Ansicht des Taskmanagers von Windows 2000 mit Prioritätsklassen

Wie man sieht, läuft der Taskmanager selber mit der Basisprioritätsklasse "Hoch". Damit hat er die Möglichkeit immer noch zu reagieren, auch wenn andere Prozesse dies nicht mehr tun. Und hier besteht auch die Gefahr, wenn man einem Prozess die Basispriorität "Echtzeit" zuweist. Dieser Prozess hätte dann eine höhere Priorität als der Taskmanager.

5.1.2 Thread-Prioritätsklassen

Windows unterstützt sieben Thread-Prioritätsklassen: "Leerlauf", "Minimum", "Niedriger als normal", "Normal", "Höher als normal", "Maximum", "Zeitkritisch". Diese Prioritäten sind relativ zur Prioritätsklasse des Prozesses definiert.

Relative Thread-Priorität	Beschreibung
Zeitkritisch	Bei geltender Prioritätsklasse "Echtzeit" arbeitet der Thread mit der Priorität 31, bei allen anderen Prioritätsklassen mit Priorität 15.
Maximum	Der Thread arbeitet mit zwei Prioritätsstufen über der Basispriorität ⁴ .
Höher als Normal	Der Thread arbeitet mit einer Prioritätsstufe über der Basispriorität.
Normal	Der Thread arbeitet mit der Basispriorität.
Niedriger als normal	Der Thread arbeitet mit einer Prioritätsstufe unterhalb der Basispriorität.
Minimum	Der Thread arbeitet mit zwei Prioritätsstufen unterhalb der Basispriorität.
Leerlauf	Bei geltender Prioritätsklasse "Echtzeit" arbeitet der Thread mit Priorität 16, bei allen anderen Prioritätsklassen mit Priorität 1.

Tabelle 10 Beschreibung der Thread-Prioritätsklassen

Die Zuordnung der prozess-bezogenen Prioritätsklassen und der relativen Thread-Prioritäten zu einer Prioritätsstufe übernimmt das Betriebssystem. Diese Zuordnung wird von Microsoft nicht festgeschrieben, damit Anwendungen unter Umständen auch lauffähig bleiben, falls Microsoft an dieser Stelle etwas ändert. Tatsächlich hat sich diese Zuordnung beim Wechsel zwischen den verschiedenen Versionen des Betriebssystems geändert.

Relative Thread-Priorität	Prioritätsklasse des Prozesses					
	Leerlauf	Niedriger als normal	Normal	Höher als normal	Hoch	Echtzeit
Zeitkritisch	15	15	15	15	15	31
Maximum	6	8	10	12	15	26
Höher als normal	5	7	9	11	14	25
Normal	4	6	8	10	13	24
Niedriger als normal	3	5	7	9	12	23
Maximum	2	4	6	8	11	22
Leerlauf	1	1	1	1	1	16

Tabelle 11 Zuordnung der prozess-bezogenen Prioritätsklassen und den relativen Thread-Prioritäten

Beispiel für die Tabelle:

Ein normaler Thread in einem Prozess mit hoher Priorität besitzt die Prioritätsstufe 13. Wird

⁴ Basispriorität: Durch die Prioritätsklasse des Prozesses festgelegte Prioritätsstufe.

nun die Prioritätsklasse des Prozesses von "Hoch" in "Leerlauf" geändert, erhält der Thread die Prioritätsstufe 4. Grund: Die Thread-Prioritäten sind relativ zur Prioritätsklasse des Prozesses definiert. Wird die Prioritätsklasse eines Prozesses geändert, hat dies keine Auswirkung auf die relative Thread-Priorität, sondern nur auf die Zuordnung zu einer Prioritätsstufe.

Die prozess-bezogene Prioritätsklasse stellt eine Abstraktion von Microsoft dar, um den internen Mechanismus der Ablaufsteuerung von der Anwendungsprogrammierung zu trennen.

5.2 Programmieren von Prioritäten

5.2.1 festlegen der Prioritätsklasse

Beim Aufruf von `CreateProcess` kann die gewünschte Prioritätsklasse über den Parameter `dwCreationFlags` festgelegt werden. Zulässige Werte sind in der folgenden Tabelle aufgelistet.

Prioritätsklasse	Konstante
Echtzeit	<code>REAL_TIME_PRIORITY_CLASS</code>
Hoch	<code>HIGH_PRIORITY_CLASS</code>
Höher als normal	<code>ABOVE_NORMAL_PRIORITY_CLASS</code>
Normal	<code>NORMAL_PRIORITY_CLASS</code>
Niedriger als normal	<code>BELOW_NORMAL_PRIORITY_CLASS</code>
Leerlauf	<code>IDLE_PRIORITY_CLASS</code>

Tabelle 12 Konstanten der Prozess-Prioritäten

Hinweis: Delphi 6 kennt die Konstanten für die Prioritätsklassen `ABOVE_NORMAL_PRIORITY_CLASS` und `BELOW_NORMAL_PRIORITY_CLASS` nicht. Grund ist der, dass diese erst ab Windows2000 verfügbar sind und von Borland in der Unit `Windows.pas` nicht berücksichtigt wurden

Abhilfe schafft eine eigene Deklaration dieser Konstanten:

```
BELOW_NORMAL_PRIORITY_CLASS = $4000;
ABOVE_NORMAL_PRIORITY_CLASS = $8000;
```

Erzeugt ein Prozess einen untergeordneten Prozess, so erbt dieser die Prioritätsklasse des Erzeuger Prozesses. Nachträglich kann die Prioritätsklasse mit der Funktion `SetPriorityClass` geändert werden.

```
function SetPriorityClass(hProcess: THandle; dwPriorityClass: DWORD): BOOL; stdcall;
```

Parameter	Bedeutung
hProcess	Handle für den Prozess
dwPriorityClass	Konstante für die Priorität

Tabelle 13 Parameter von *SetPriorityClass*

Das Gegenstück dazu ist die Funktion *GetPriorityClass*, welchen die Prioritätsklasse von dem mit *hProcess* angegebenen Prozess zurück gibt.

```
function GetPriorityClass(hProcess: THandle): DWORD; stdcall;
```

Parameter	Bedeutung
hProcess	Handle für den Prozess

Tabelle 14 Parameter von *GetPriorityClass*

Rückgabewert ist die Priorität als eine Konstante aus obiger Tabelle.

5.2.2 Festlegen der Thread-Priorität

Da *CreateThread* keine Möglichkeit bietet die Thread-Priorität festzulegen, wird jeder neu erzeugte Thread mit der relativen Priorität "Normal" erzeugt. Die relative Thread-Priorität kann aber mit der Funktion *SetThreadPriority* festgelegt werden.

```
function SetThreadPriority(hThread: THandle; nPriority: Integer): BOOL; stdcall;
```

Parameter	Bedeutung
hThread	Handle des Threads
nPriority	Priorität (Ein Wert aus folgender Tabelle)

Tabelle 15 Parameter von *SetThreadPriority*

Relative Thread-Priorität	Konstante
Zeitkritisch	THREAD_PRIORITY_TIME_CRITICAL
Maximum	THREAD_PRIORITY_HIGHEST
Höher als normal	THREAD_PRIORITY_ABOVE_NORMAL
Normal	THREAD_PRIORITY_NORMAL
Niedriger als normal	THREAD_PRIORITY_BELOW_NORMAL
Minimum	THREAD_PRIORITY_LOWEST
Leerlauf	THREAD_PRIORITY_IDLE

Tabelle 16 Thread-Prioritäts Konstanten

Will man nun einen Thread mit einer anderen Priorität erzeugen, so erzeugt man ihn mit `CreateThread` (oder dem Wrapper `BeginThread`) im angehaltenen Zustand (`CREATE_SUSPENDED`), setzt mit `SetThreadPriority` die Priorität und ruft dann `ResumeThread` auf, um ihn zuteilungsfähig zu machen.

Mit dem Gegenstück zu `SetThreadPriority`, `GetThreadPriority`, lässt sich die aktuelle Thread-Priorität ermitteln.

```
function GetThreadPriority(hThread: THandle): Integer; stdcall;
```

Parameter	Bedeutung
hThread	Handle des Threads

Tabelle 17 Parameter von `GetThreadPriority`

Rückgabewert ist die Priorität als eine Konstante aus Tabelle 16.

Hinweis: Windows stellt keine Funktion zur Verfügung, mit der sich die Prioritätsstufe eines Threads ermitteln lässt. Microsoft behält es sich vor den Algorithmus zur Ermittlung der Prioritätsstufe zu ändern. Es sollte also keine Anwendung entworfen werden, die eine Kenntnis darüber erfordert. Eine Beschränkung auf die Prioritätsklasse und der relativen Thread-Priorität sollte sicherstellen, dass die Anwendung auch unter zukünftigen Windows-Versionen lauffähig ist.

5.2.3 Prioritätsanhebung durch das System

Wie schon gesagt, ermittelt das System die Prioritätsstufe eines Threads durch die Kombination der Prioritätsklasse des übergeordneten Prozesses mit der relativen Thread-Priorität. Es kann aber die Prioritätsstufe dynamisch anheben in Folge eines E/A Ereignisses, Einreihen einer Fensternachricht in die Warteschlange oder bei einem Zugriff auf die Festplatte.

Beispiel:

Ein Thread mit der Priorität "Normal", der in einem Prozess mit der Basispriorität "Hoch" läuft hat die Prioritätsstufe 13. In Folge eines Tastendrucks, welcher dazu führt, dass eine `WM_KEYDOWN` Nachricht in die 'Warteschlange eingereicht wird, wird der Thread ist nun zuteilungsfähig und der Tastaturtreiber kann das System veranlassen die Prioritätsstufe dieses Threads um zwei Stufen zu erhöhen. Für die Dauer der nächsten Zeitscheibe läuft dieser Thread also mit der Prioritätsstufe 15. Über die nächsten Zeitscheiben hinweg erfolgt dann eine stufenweise Absenkung der Prioritätsstufe um jeweils eins, bis der Thread wieder bei seiner Basisprioritätsstufe angekommen ist.

Es ist zu beachten, dass die Prioritätsstufe niemals unter der Basisprioritätsstufe abgesenkt werden kann. Des weiteren behält es sich Microsoft vor, dieses Konzept in Zukunft zu ändern / zu verbessern, um ein optimales Reaktionsverhalten des Gesamtsystems zu erzielen. Aus diesem Grund ist es von Microsoft auch nicht dokumentiert.

Eine dynamische Prioritätsanhebung erfolgt nur bei Threads mit einer Prioritätsstufe zwischen

1 und 15. Deshalb bezeichnet man diese Prioritätsstufen auch als dynamische Prioritäten. Es erfolgt auch nie eine dynamische Anhebung in den Echtzeitbereich, da viele Systemprozesse in diesem Bereich laufen und so verhindert wird, dass diese Threads mit dem Betriebssystem konkurrieren.

Diese dynamische Anhebung kann, laut einiger Entwickler, zu einem nachteiligen Leistungsverhalten der eigenen Anwendung führen. Microsoft hat daraufhin zwei Funktionen eingeführt, um die dynamische Prioritätsstufenanhebung zu deaktivieren.

```
function SetProcessPriorityBoost(hThread: THandle; DisablePriorityBoost: Bool): BOOL;
stdcall;
function SetThreadPriorityBoost(hThread: THandle; DisablePriorityBoost: Bool): BOOL;
stdcall;
```

Parameter	Bedeutung
hThread	Handle des Prozesses / Threads
DisablePriorityBoost	Aktivieren oder deaktivieren der dynamischen Prioritätsanhebung

Tabelle 18 Parameter von *SetProcessPriorityBoost* / *SetThreadPriorityBoost*

Mit den Gegenstücken

```
function GetProcessPriorityBoost(hThread: THandle; var DisablePriorityBoost: Bool): BOOL;
stdcall;
function GetThreadPriorityBoost(hThread: THandle; var DisablePriorityBoost: Bool): BOOL;
stdcall;
```

Parameter	Bedeutung
hThread	Handle des Prozesses / Threads
var DisablePriority-Boost	Boolsche Variable die den Status aufnimmt

Tabelle 19 Parameter von *GetProcessPriorityBoost* / *GetThreadPriorityBoost*

lässt sich der Status (aktiv / inaktiv) der dynamischen Prioritätsanhebung abfragen.

Es gibt noch eine weitere Situation in der eine dynamische Prioritätsstufenanhebung erfolgen kann. Gegeben sei folgendes Szenario: Thread A läuft mit der Priorität von 4. Im System existiert noch ein Thread B mit der Prioritätsstufe 8, der ständig zuteilungsfähig ist. In dieser Konstellation würde Thread A nie Rechenzeit zugeteilt bekommen. Bemerkt das System, dass ein Thread über drei bis vier Sekunden derart blockiert wird, setzt es die Prioritätsstufe des Threads für zwei aufeinander folgende Zeitscheiben auf die Stufe 15 und senkt sie danach sofort wieder die Basisprioritätsstufe ab.

5.3 Beispielanwendung "Priority-Demo"

Anhand dieser Beispielanwendung kann mit den Prioritäten experimentiert werden. Es empfiehlt sich, zwei Instanzen des Programms zu starten und den Taskmanager nebenbei zu öffnen, um zu beobachten, wie sich die verschiedenen Kombinationen der Prioritäten auf das Verhalten der Anwendungen auswirken.




Abbildung 4 Beispielanwendung "Priority-Demo"

Ein Klick auf den Button "Suspend" erzeugt einen zweiten Thread, welcher einfach nur eine MessageBox anzeigt und den primären Thread lahmlegt. Dadurch kann das Hauptfenster keine Fensternachrichten mehr verarbeiten, auch kein WM_PAINT, was durch Verschieben der MessageBox deutlich wird. Die macht hinreichend deutlich, dass sich der primäre Thread im angehaltenen Zustand befindet.

Anmerkung:

Seltsamerweise scheint die MessageBox nicht angezeigt zu werden, wenn das Programm aus der IDE gestartet wird. Startet man die Anwendung direkt aus dem Explorer, funktioniert sie wie gewünscht.

6 Thread-Synchronisation

Im System haben alle Threads Zugang zu allen Systemressourcen, ansonsten könnten sie ihre Arbeit nicht verrichten. Da es aber nun nicht sein darf, dass jeder Thread, zu jeder Zeit, jede Ressource manipulieren kann – während ein Thread eine Ressource liest, darf kein anderer Thread im gleichen Moment schreibend auf die Ressource / Speicherblock zugreifen – muss der Zugriff "synchronisiert" werden. Die Synchronisation im Benutzermodus kann entweder über einen atomaren Variablenzugriff mit den Interlocked-Funktionen erfolgen, wenn es sich um einen 32-Bit-Wert handelt oder über kritische Abschnitte, wenn es komplexe Datenstrukturen sind.

6.1 Atomarer Variablenzugriff

Unter atomaren Zugriffen versteht man, dass ein Thread mit der Gewähr auf eine Ressource zugreifen kann, ohne dass eine anderer Thread gleichzeitig auf sie zugreift. Ein Beispiel:

```
var
  g_Int: Integer = 0;
function ThreadOne(p: Pointer): Integer;
begin
  Inc(g_Int);
  result := 0;
end;

function ThreadTwo(p: Pointer): Integer;
begin
  Inc(g_Int);
  result := 0;
end;
```

Hier inkrementieren zwei Threads eine globale Integer-Variable um 1. Nun, was erwartet man, wenn beide Threads die Variable bearbeitet haben? Dass sie den Wert 2 hat, da sie ja in beiden Threads um 1 erhöht wurde. Aber muss das so sein? Nehmen wir an der Compiler generiert folgenden Code:

```
mov eax, [g_Int]    // Wert von g_Int in Register kopieren
inc eax            // Wert im Register inkrementieren
mov [g_Int], eax   // Neuen Wert wieder in g_Int speichern
```

Führen nun beide Threads ihren Code nacheinander aus, passiert folgendes:

```
mov eax, [g_Int]    // Thread 1: Kopiert 0 in Register
inc eax            // Thread 1: Inkrementiert Registerinhalt um 1
mov [g_Int], eax   // Thread 1: Speichert 1 in g_Int
mov eax, [g_Int]    // Thread 2: Kopiert 1 in Register
inc eax            // Thread 2: Inkrementiert Registerinhalt um 1
mov [g_Int], eax   // Thread 2: Speichert 2 in g_Int
```

Das Ergebnis ist wie erwartet 2. Aber muss das so ablaufen? Windows ist ja ein präemptives Multitasking Betriebssystem, es ist also nicht gewährleistet, dass beide Threads schön nacheinander durchlaufen. Wie könnte es aussehen, wenn ein Thread unterbrochen wird?

```

mov eax, [g_Int]    // Thread 1: Kopiert 0 in Register
inc eax            // Thread 1: Inkrementiert Registerinhalt um 1

mov eax, [g_Int]    // Thread 2: Kopiert 0 in Register
inc eax            // Thread 2: Inkrementiert Registerinhalt um 1
mov [g_Int], eax    // Thread 2: Speichert 1 in g_Int

mov [g_Int], eax    // Thread 1: Speichert 1 in g_Int

```

Und schon lautet unser Ergebnis nicht mehr 2 sondern 1! Das Ergebnis hängt von vielerlei Gegebenheiten ab: Wie wird der Code vom Compiler generiert? Welcher Prozessor führt den Code aus? Wie viele Prozessoren gibt es im System? Windows wäre als Multitasking Betriebssystem nicht zu gebrauchen, wenn es für unser Problem keine Lösung bereitstellen würde.

Und zwar stellt Windows die Interlocked-Funktionen bereit. Alle Funktionen dieser Familie haben eins gemeinsam, sie bearbeiten einen Wert auf atomare Weise. Nehmen wir als Beispiel mal die Funktion `InterlockedExchangeAdd()` (weitere Funktionen der Interlocked-Familie finden sich in den Demos: `InterlockedExchangeAdd_Demo` und `SpinLock`):

```

function InterlockedExchangeAdd(Addend: PLongint; Value: Longint): Longint stdcall; overload;
function InterlockedExchangeAdd(var Addend: Longint; Value: Longint): Longint stdcall; overload;

```

Parameter	Bedeutung
Addend (PLongint)	Zeiger auf ein Integer, der den neuen Wert aufnimmt.
Addend (Longint)	Variable, die den neuen Wert aufnimmt.
Value	Wert, um den inkrementiert werden soll.

Tabelle 20 Parameter von `InterlockedExchangeAdd`

`InterlockedExchangeAdd` inkrementiert einfach den ersten Parameter um den in `Value` angegebenen Wert und stellt sicher, dass der Zugriff atomar erfolgt. Unseren obigen Code können wir also wie folgt umschreiben:

```

function ThreadOne(p: Pointer): Integer;
begin
  InterlockedExchangeAdd(g_Int, 1);
  result := 0;
end;

function ThreadTwo(p: Pointer): Integer;
begin
  InterlockedExchangeAdd(g_Int, 1);
  result := 0;
end;

```

und so sicherstellen, dass unsere globale Variable `g_Int` korrekt und wie erwartet inkrementiert wird. Wichtig dabei ist, dass alle beteiligten Threads die Änderung an der Variablen nur atomar vornehmen!

Hinweis: Die Interlocked-Funktionen sind zu dem auch noch sehr schnell. Ein Aufruf einer Interlocked-Funktion verbraucht während ihrer Ausführung nur einige CPU-Befehlszyklen (in der

Regel weniger als 50) und es findet kein Wechsel vom Benutzer- in den Kernel-Modus statt, was gewöhnlich mehr als 1000 Zyklen benötigt.

Werfen wir noch einen Blick auf die Funktion `InterlockedExchange()`:

```
function InterlockedExchange(var Target: Integer; Value: Integer): Integer; stdcall;
```

Parameter	Bedeutung
var Target	Variable, die den Ziel-Wert aufnimmt
Value	Wert für das Ziel

Tabelle 21 Parameter von `InterlockedExchange`

Diese Funktion ersetzt den Wert, der in `Target` steht, durch den Wert der in `Value` steht. Das besondere ist, dass `InterlockedExchange` den ursprünglichen Wert von `Target` als Funktions-Ergebnis zurückliefert. Damit lässt sich ein so genanntes `SpinLock` implementieren:

```
var
  g_IsInUse: Integer = 1; // Status Flag; 1: in Gebrauch, 0: frei

function WatchDogThread(p: Pointer): Integer;
begin
  // läuft so lange g_IsInUse 1 ist
  while InterlockedExchange(g_IsInUse, 1) <> 0 do
    Sleep(0);
  // g_IsInUse ist 0 geworden, Schleife wurde verlassen
  writeln('frei.');
```

Die `while`-Schleife wird immer wieder durchlaufen, wobei sie den Wert von `g_IsInUse` auf 1 setzt und prüft, ob er vorher ungleich 0 war (1 bedeutet Ressource ist in Gebrauch, 0 auf Ressource kann zugegriffen werden). Ein anderer Thread bearbeitet die zu überwachende Ressource und setzt `g_IsInUse` auf 0, wenn er damit fertig ist. Damit liefert die `while`-Schleife irgendwann einmal `False`, die Ressource ist frei und wird verlassen. Dieses Vorgehen bezeichnet man als `SpinLock`.

Diese Technik ist allerdings nicht sehr effizient, da ein `SpinLock` viel Rechenzeit verbraucht, denn es muss ja ständig ein Wert verglichen werden, der auf "wundersame" Weise an anderer Stelle geändert wird. Außerdem setzt dieser Code voraus, dass alle Threads, die mit dem `SpinLock` arbeiten auf der gleichen Prioritätsstufe laufen.

Hinweis: Es gibt keine `Interlocked`-Funktion, die nur einen Wert liest, da eine solche Funktion überflüssig ist. Wenn ein Thread einen Wert liest, deren Inhalt mit einer `Interlocked`-Funktion geändert wurde, handelt es sich immer um einen brauchbaren Wert.

6.2 Kritische Abschnitte

Handelt es sich bei den Ressourcen nicht um reine 32-Bit Werte, dann kommt man nicht mehr mit den `Interlocked`-Funktionen aus - hier helfen jetzt kritische Abschnitte. Ein kritischer Ab-

schnitt stellt einen Bereich innerhalb des Programmcodes dar, der bei seiner Ausführung exklusiven Zugriff auf eine oder mehrere Ressourcen benötigt. Hier mal ein Beispiel, wie kritische Abschnitte eingesetzt werden:

```
var
  g_Index: Integer = 0;
  g_cs: RTL_CRITICAL_SECTION;
  F: TextFile;
function FirstThread(p: Pointer): Integer;
begin
  result := 0;
  while g_Index < MAX_TIMES do
  begin
    // in CriticalSection eintreten, Ressource sperren
    EnterCriticalSection(g_cs);
    writeln(F, IntToStr(GetTickCount()));
    Inc(g_Index);
    // CriticalSection verlassen
    LeaveCriticalSection(g_cs);
    Sleep(0);
  end;
end;

function SecondThread(p: Pointer): Integer;
begin
  result := 0;
  while g_Index < MAX_TIMES do
  begin
    // in CriticalSection eintreten, Ressource sperren
    EnterCriticalSection(g_cs);
    Inc(g_Index);
    writeln(F, IntToStr(GetTickCount()));
    // CriticalSection verlassen
    LeaveCriticalSection(g_cs);
    Sleep(0);
  end;
end;
```

Wie man sieht greifen beide Threads auf die gleiche Datei zu (F) und schreiben einen Wert (GetTickCount()) in sie hinein, gleichzeitig wird eine globale Variable (g_Index) erhöht, welche für die Abbruchbedingung benötigt wird. Damit beim Schreiben nichts durcheinander geht, sind in den Threads die Zugriffe auf die Datei und die globale Variable mit kritischen Abschnitten geschützt.

Wie funktionieren kritische Abschnitte nun? Es wurde eine Struktur des Typs CRITICAL_SECTION mit dem Namen g_cs angelegt und jede Befehlsfolge, die auf gemeinsame Ressourcen zugreift wird in einen Aufruf von EnterCriticalSection und LeaveCriticalSection geklammert, denen man einen Zeiger auf die CRITICAL_SECTION-Struktur als Parameter übergibt. Das Prinzip ist nun folgendes: Tritt nun ein Thread in einen kritischen Abschnitt ein oder versucht es, überprüft es mittels der CRITICAL_SECTION-Struktur, ob nicht schon auf die Ressource / den Code-Abschnitt zugegriffen wird. Ist die Ressource "frei", betritt es den kritischen Abschnitt und hängt ein "Besetzt"-Schild an die Tür. Hat es die Ressourcen manipuliert und verlässt den kritischen Abschnitt wieder, wird das "Besetzt"-Schild praktisch wieder mit LeaveCriticalSection umgedreht, so dass es jetzt wieder "frei" zeigt.

Wichtig ist, wie auch bei den Interlocked-Funktionen, dass jeder Thread, der auf gemeinsame Ressourcen zugreift, dies nur innerhalb eines kritischen Abschnittes tut. Ein Thread, der dies nicht tut, missachtet quasi das "Besetzt"-Schild – er prüft es gar nicht erst – und könnte so die Ressourcen kompromittieren. Andersherum, vergisst ein Thread nach Manipulation der Ressourcen ein LeaveCriticalSection aufzurufen, ähnelt dies dem Verhalten beim Verlassen das "Besetzt"-Schild nicht wieder umzudrehen und so zu signalisieren, dass die Ressourcen wieder "frei" sind.

6.2.1 Wie funktionieren kritische Abschnitte

Wie funktionieren jetzt aber kritische Abschnitte? Grundlage ist die Datenstruktur `CRITICAL_SECTIONS`. Sie ist nicht dokumentiert, da Microsoft der Meinung ist, dass sie vom Programmierer nicht verstanden werden müsste. Dies ist auch nicht nötig, da man API-Funktionen aufruft, um sie zu manipulieren. Diese Funktionen kennen die Details, die zur Manipulation nötig sind und garantieren so den konsistenten Zustand der Struktur.

In der Regel werden Variablen dieser Struktur global angelegt damit alle Threads auf sie zugreifen können. Des weiteren ist es nötig, dass die Elemente der Struktur initialisiert werden. Dies geschieht mit dem Aufruf von `InitializeCriticalSection()`:

```
procedure InitializeCriticalSection(var lpCriticalSection: TRTLCriticalSection); stdcall;
```

Parameter	Bedeutung
var lpCriticalSection	Zeiger auf eine <code>RTL_CRITICAL_SECTION</code> Struktur.

Tabelle 22 Parameter von `InitializeCriticalSection`

Sobald die kritischen Abschnitte nicht mehr gebraucht werden, sollte die `CRITICAL_SECTION`-Struktur zurückgesetzt werden mit dem Aufruf von `DeleteCriticalSection`:

```
procedure DeleteCriticalSection(var lpCriticalSection: TRTLCriticalSection); stdcall;
```

Parameter	Bedeutung
var lpCriticalSection	Zeiger auf eine <code>RTL_CRITICAL_SECTION</code> Struktur.

Tabelle 23 Parameter von `DeleteCriticalSection`

Wie schon im Beispiel zu sehen war, muss jedem Codestück, welches auf eine gemeinsame Ressource zugreift ein Aufruf von `EnterCriticalSection` vorangehen.

```
procedure EnterCriticalSection(var lpCriticalSection: TRTLCriticalSection); stdcall;
```

Parameter	Bedeutung
var lpCriticalSection	Zeiger auf eine <code>RTL_CRITICAL_SECTION</code> Struktur.

Tabelle 24 Parameter von `EnterCriticalSection`

`EnterCriticalSection` untersucht dabei die Elementvariablen der `CRITICAL_SECTION`-Struktur und entnimmt ihnen den Zustand der Ressourcen. Dabei geschieht folgendes:

- Greift gerade kein Thread auf die Ressource zu, aktualisiert `EnterCriticalSection` die Elemente entsprechend, um anzuzeigen, dass dem aufrufenden Thread Zugriff gewährt wurde.
- Zeigt das Element an, dass der aufrufende Thread schon Zugang zu den Ressourcen hat, wird der Zähler erhöht, der festhält, wie oft dem aufrufenden Thread schon Zugriff gewährt wurde.

- Stellt die Funktion EnterCriticalSection fest, dass die Ressource sich im Zugriff eines anderen Threads befindet, überführt EnterCriticalSection den aufrufenden Thread in den Wartezustand. Das System merkt sich das und sorgt dafür dass der aufrufende Thread sofort zuteilungsfähig wird, sobald die Ressourcen wieder freigegeben werden.

Am Ende eines Codestücks, das auf eine gemeinsame Ressource zugreift, muss die Funktion LeaveCriticalSection aufgerufen werden.

```
procedure LeaveCriticalSection(var lpCriticalSection: RTL_CRITICAL_SECTION); stdcall;
```

Paramter	Bedeutung
var lpCriticalSection	Zeiger auf eine RTL_CRITICAL_SECTION Struktur.

Tabelle 25 Parameter von LeaveCriticalSection

Die Funktion decremientiert einen Zähler, der anzeigt wie oft dem aufrufenden Thread der Zugriff auf die Ressource gewährt wurde. Fällt der Zähler auf 0, prüft die Funktion, ob weitere Threads auf die Freigabe dieses kritischen Abschnittes warten. Ist dies der Fall überführt es einen wartenden Thread in den zuteilungsfähigen Zustand. Gibt es keine wartenden Threads, aktualisiert LeaveCriticalSection das Element ebenfalls, um anzuzeigen, dass im Moment kein Thread auf die Ressourcen zugreift.

6.2.2 Kritische Abschnitte und SpinLocks

Wird ein Thread in den Wartezustand überführt, hat dies zur Folge, dass er von den Benutzer-Modus in den Kernel-Modus überwechseln muss. Dies nimmt etwa 1000 CPU-Befehlszyklen in Anspruch. Um nun die Effizienz von kritischen Abschnitten zu verbessern kann man den kritischen Abschnitt mit InitializeCriticalSectionAndSpinCount initialisieren:

```
function InitializeCriticalSectionAndSpinCount(var lpCriticalSection: RTL_CRITICAL_SECTION; dwSpinCount: DWORD): BOOL; stdcall;
```

Parameter	Bedeutung
var lpCriticalSection	Zeiger auf eine RTL_CRITICAL_SECTION Struktur.
dwSpinCount	Anzahl für die Schleifendurchläufe für die Abfrage des SpinLocks.

Tabelle 26 Parameter von InitializeCriticalSectionAndSpinCount

Der Parameter dwSpinCount gibt dabei die Anzahl der Schleifendurchläufe der Abfrage des SpinLocks an. Also wie oft der Thread versuchen soll Zugriff auf die Ressourcen zu bekommen bevor er in den Wartezustand geht und so in den Kernel-Modus wechselt. Erfolgt der Aufruf auf einem Einzelprozessorsystem wird dieser Parameter ignoriert und auf 0 gesetzt, da es nutzlos ist in diesem Fall einen SpinLockzähler zusetzen: Der Thread, der die Ressource kontrolliert, kann die Kontrolle nicht abgeben, solange ein anderer Thread aktiv wartet, also das SpinLock testet.

Der SpinCount eines kritischen Abschnittes kann mit SetCriticalSectionSpinCount nachträglich geändert werden:

```
function SetCriticalSectionSpinCount(var lpCriticalSection: TRTLCriticalSection;  
    dwSpinCount: DWORD): DWORD; stdcall;
```

Parameter	Bedeutung
var lpCriticalSection	Zeiger auf eine RTL_CRITICAL_SECTION Struktur.
dwSpinCount	Anzahl für die Schleifendurchläufe für die Abfrage des SpinLocks.

Tabelle 27 Parameter von SetCriticalSectionSpinCount

→ Demo: InterLockedExchangeAdd, CriticalSection, SpinLock

7 Thread-Synchronisation mit Kernel-Objekten

Im vorherigen Kapitel habe ich die Thread-Synchronisation im Benutzermodus besprochen. Ihr großer Vorteil ist, dass sie sich positiv auf das Leistungsverhalten auswirken. Bei der Synchronisation mit Kernelobjekten ist es nötig, dass der aufrufende Thread vom Benutzermodus in den Kernelmodus wechselt. Dies ist jedoch mit nicht unerheblichen Rechenaufwand verbunden. So braucht auf der x86-Plattform die COU ca. 1.000 Befehlszyklen für jede Richtung des Moduswechsels. Die Thread Synchronisation im Benutzermodus hat jedoch den Nachteil, dass sie gewissen Einschränkungen unterliegt und einfach eine Funktionalität nicht bietet, die man aber das eine oder andere mal braucht. Beispiele hierzu findet man in den Demos (→Demo: WaitForSingleObject, →Demo: HandShake, →Demo: WaitForMultipleObject) und in den folgenden Unterkapiteln.

Hier soll es nun um die Thread-Synchronisation mit Kernelobjekten gehen. Was Kernelobjekte sind habe ich weiter oben schon mal erläutert. Im Prinzip geht es in diesem Kapitel darum eben diese Kernelobjekte, in diesem Fall beschränke ich mich des Themas willen hauptsächlich auf Thread- und Prozessobjekte, miteinander zu synchronisieren.

Hinsichtlich der Thread-Synchronisation wird bei jedem Kernelobjekt zwischen dem Zustand *signalisiert* und *nicht-signalisiert* unterschieden. Nach welchen Regeln dies geschieht ist in Windows für jeden Objekttypen eigens festgelegt. So werden zum Beispiel Prozessobjekte immer im Zustand nicht-signalisiert erzeugt und gehen in den Zustand signalisiert über, wenn der Prozess endet. Diesen Zustandswechsel übernimmt das Betriebssystem automatisch. Innerhalb des Prozessobjektes gibt es einen bool'schen Wert, der beim Erzeugen des Objektes mit FALSE (nicht-signalisiert) initialisiert wird. Bei der Beendigung des Prozesses ändert das Betriebssystem automatisch diesen bool'schen Wert und setzt ihn auf TRUE, um den signalisierten Zustand des Objektes anzuzeigen. Soll also Code erstellt werden, der prüft, ob ein bestimmter Prozess noch aktiv ist, braucht nur, mit der dazu zur Verfügung gestellten Funktion, dieser Wert abgefragt zu werden.

Im speziellen werde ich in diesem Kapitel darauf eingehen, wie es einem Thread ermöglicht werden kann, darauf zu warten, dass ein bestimmtes Kernel-Objekt signalisiert wird und wie so etwas zur Thread-Synchronisation genutzt werden kann

7.1 WaitForxxx-Funktionen

Die *Wait*-Funktionen veranlassen einen Thread, sich freiwillig so lange in den Ruhezustand zu begeben, bis ein bestimmtes Kernel-Objekt signalisiert wird. Während der Thread wartet braucht er keine Rechenzeit, das Warten ist also äußerst ressourcenschonend.

Die wohl am meisten gebrauchte Wait-Funktion ist wohl *WaitForSingleObject*:

```
function WaitForSingleObject(hHandle: THandle; dwMilliseconds: DWORD): DWORD; stdcall;
```

Parameter	Bedeutung
hHandle	Handle des Kernel-Objektes auf das gewartet werden soll
dwMilliseconds	Zeit in Millisekunden die gewartet werden soll (INFINITE = keine zeitliche Begrenzung)
Rückgabewerte	WAIT_OBJECT_0: Objekt wurde signalisiert WAIT_TIMEOUT: Zeitüberschreitung WAIT_FAILED: Fehler beim Aufruf der Funktion (ungültiges Handle oder ähnliches)

Im Demo-Programm "wfso" im Unterordner "WaitForSingleObject" habe ich die Benutzung von WaitForSingleObject demonstriert:

```

program wfso;

{$APPTYPE CONSOLE}

uses
  Windows;

function Thread(p: Pointer): Integer;
resourcestring
  rsReturn = 'Druecken Sie die Eingabetaste - oder auch nicht...';
begin
  Writeln(rsReturn);
  Readln;
  result := 0;
end;

var
  hThread: THandle;
  ThreadID: Cardinal;
  wf: DWORD;
resourcestring
  rsThreadStart = 'Thread wird gestartet!';
  rsWO = 'Thread wurde vom Benutzer beendet!';
  rsWT = 'Thread wurde nicht innerhalb von 10 Sekunden beendet!';
  rsWF = 'Fehler!';
begin
  Writeln(rsThreadStart);
  Writeln('');
  hThread := BeginThread(nil, 0, @Thread, nil, 0, ThreadID);
  if hThread <> INVALID_HANDLE_VALUE then
    begin
      wf := WaitForSingleObject(hThread, 5000);
      case wf of
        WAIT_OBJECT_0: Writeln(rsWO);
        WAIT_TIMEOUT: Writeln(rsWT);
        WAIT_FAILED: Writeln(rsWT);
      end;
      CloseHandle(hThread);
    end;
end.

```

Im Hauptprogramm wird ein Thread abgespalten, den der Benutzer durch Drücken der Eingabetaste beenden kann. Das Hauptprogramm wartet mit WaitForSingleObject auf die Signalisierung des abgespaltenen Threads und zwar genau fünf Sekunden lang. Oder anders ausgedrückt: Der Aufruf von WaitForSingleObject weist das Betriebssystem an, den Thread so lange nicht als zuteilungsfähig zu behandeln, bis entweder der Thread signalisiert oder die fünf Sekunden überschritten wurden. Dann wertet es den Rückgabewert von WaitForSingleObject aus und gibt eine entsprechende Meldung in der Konsole aus. Betätigt der Benutzer die Eingabetaste wird der Thread normal beendet und das Thread-Objekt wird signalisiert; Rückgabewert ist WAIT_OBJECT_0. Macht der Benutzer gar nichts, bricht nach fünf Sekunden die Funktion WaitForSingleObject das Warten ab und kehrt mit WAIT_TIMEOUT zurück. Sollte ein Fehler auftreten, weil zum Beispiel das Handle ungültig war, gibt die Funktion WAIT_FAILED zurück.

→ Demo: WaitForSingleObject

Eine praktische Anwendung wäre, wenn zum Beispiel in einem Thread etwas initialisiert wird und der Hauptthread warten muss bis der abgespaltene Thread seine Arbeit getan hat.

Des Weiteren stellt Windows noch eine Funktion bereit, um den Signalisierungssatus mehrerer Kernel-Objekte gleichzeitig abzufragen: *WaitForMultipleObjects*.

→ Demo: WaitForMultipleObjects

7.2 Ereignisobjekte

Ereignisse finden meist dann Verwendung, wenn ein Thread signalisieren will, dass es mit etwas fertig ist oder mit etwas begonnen hat und andere Threads über seinen Zustand informieren will.

Mit der Funktion *CreateEvent* erzeugt man ein Ereignisobjekt:

```
function CreateEvent(lpEventAttributes: PSecurityAttributes;
    bManualReset, bInitialState: BOOL; lpName: PChar): THandle; stdcall;
```

Parameter	Bedeutung
lpEventAttributes	Zeiger auf eine Sicherheitsattributstruktur
bManuelReset	Manuel-Reset (True) oder Auto-Reset
bInitialState	Erzeugung des Objektes im signalisierten Zustand oder nicht
lpName	Dient zur Identifizierung des Objektes

Als Rückgabewert gibt *CreateEvent* ein prozessbezogenes Handle auf das Ereignisobjekt zurück.

Hat man einen Namen für das Ereignisobjekt vergeben, so kann man wieder Zugriff auf selbiges erlangen mit der Funktion: *OpenEvent*:

```
function OpenEvent(dwDesiredAccess: DWORD; bInheritHandle: BOOL; lpName: PChar): THandle;
stdcall;
```

Parameter	Bedeutung
dwDesiredAccess	Sicherheitsbeschreiber für das Objekt
bInheritHandle	Soll das Handle vererbt werden oder nicht
lpName	Name des Events

Wie üblich sollte die Funktion *CloseHandle* aufgerufen werden, wenn das Objekt nicht mehr länger gebraucht wird.

So bald das Ereignisobjekt erzeugt wurde, kann man dessen Zustand mit den Funktionen *SetEvent* und *ResetEvent* kontrollieren:

```
function SetEvent(hEvent: THandle): BOOL; stdcall;
```

```
function ResetEvent(hEvent: THandle): BOOL; stdcall;
```

Als Parameter erwarten beide Funktionen jeweils ein gültiges Handle auf ein Ereignisobjekt. *SetEvent* ändert den Zustand des Ereignisobjektes in signalisiert und *ResetEvent* ändert ihn wieder in nicht-signalisiert.

Zusätzlich hat Microsoft noch einen Nebeneffekt für erfolgreiches Warten für Auto-Reset-Ereignisse definiert: Ein Auto-Reset-Ereignisobjekt wird automatisch in den Zustand nicht-signalisiert zurückgesetzt, wenn ein Thread erfolgreich auf das Ereignis wartet. Es ist deshalb

nicht nötig, extra noch *ResetEvent* für ein Auto-Reset-Ereignis aufzurufen. Im Gegensatz dazu hat Microsoft für ein manuelle Ereignisse keinen solchen Nebeneffekt definiert.

8 Das VCL Thread-Objekt

Die VCL bietet eine recht einfache Möglichkeit, mit Threads zu arbeiten und zwar die Klasse TThread. Sie übernimmt für einen eigentlich fast alles, was man zur Thread Programmierung braucht und kann sie dazu handhaben wie jedes beliebige Objekt in Delphi auch, zum Beispiel eine Stringliste oder ähnliches.

Die Klasse TThread stellt dazu ihre Eigenschaften und Methoden bereit. An dieser Stelle sollen nur die wichtigsten kurz angesprochen werden.

8.1 Erzeugen, Eigenschaften, Methoden

8.1.1 Erzeugen

Den Code für ein TThread-Objekt lässt sich mit Hilfe des Assistenten der IDE recht schnell erledigen: Daten | Neu | weitere | TThread-Objekt. Nach Eingabe des Klassennames, erstellt einem die IDE ein Grundgerüst:

```
unit Unit3;

interface

uses
  Classes;

type
  TMyFirstThread = class(TThread)
  private
    { Private-Deklarationen }
  protected
    procedure Execute; override;
  end;

implementation

{ Wichtig: Methoden und Eigenschaften von Objekten in VCL oder CLX können
  nur in Methoden verwendet werden, die Synchronize aufrufen, z.B.:

  Synchronize(UpdateCaption);

  wobei UpdateCaption so aussehen könnte:

  procedure TMyFirstThread.UpdateCaption;
  begin
    Form1.Caption := 'In einem Thread aktualisiert';
  end; }

{ TMyFirstThread }

procedure TMyFirstThread.Execute;
begin
  { Thread-Code hier einfügen }
end;

end.
```

Und eigentlich wird das wichtigste, was man nun über die Klasse wissen muss, um sie korrekt zu benutzen, schon im Kommentar erklärt.

1. Jeder Zugriff auf Elemente der VCL muss mit der Methode Synchronize synchronisiert werden.
2. Der Code welchen der Thread ausführen soll gehört in die Execute Methode.

Erzeugen einer Instanz der Klasse TThread

Erzeugt wird eine Instanz des der Thread Klasse wie bei jedem Objekt mit der Methode Create. Create übernimmt einen Parameter CreateSuspended, so lange man sie nicht überschrieben hat, was auch möglich ist. Dieser eine Parameter gibt an, ob der Thread im angehaltenen (True) oder im zuteilungsfähigen (False) Zustand erzeugt werden soll. Will man noch den Feld-Variablen des Threads Werte zuweisen und oder seine Priorität setzen, ist es sinnvoll, ihn im angehaltenen Zustand zu erzeugen, alles zu initialisieren und ihn dann in den zuteilungsfähigen Zustand zu versetzen.

Parameter an den Thread zu übergeben ist auch etwas einfacher. Man muss nicht den Umweg über einen Zeiger auf eine eigene Datenstruktur gehen, sondern man legt einfach in der Thread-Klasse Felder im public Abschnitt an:

```
type
  TMyThreads = class(TThread)
  private
    { Private-Deklarationen }
  protected
    procedure Execute; override;
  public
    // Zeilenindex des Listviews
    FIndex: Integer;
  end;
```

Und dann kann man, nachdem man das Thread-Objekt erzeugt hat, ihnen ganz einfach Werte zuweisen:

```
ThreadArray[Loop] := TMyThreads.Create(True);
ThreadArray[Loop].FIndex := Loop;
```

8.1.2 Eigenschaften

Eigenschaft	Erklärung
FreeOnTerminate	Zerstört das Thread Objekt automatisch, wenn es seinen Code fertig ausgeführt hat.
Handle	Handle des Threads.
Priority	Thread Priorität (tpIdle, tpLowest, tpLower, tpNormal, tpHigher, tpHighest, tpTimeCritical).

Eigenschaft	Erklärung
Suspended	Zeigt an, ob ein Thread sich im angehaltenen Zustand befindet oder nicht.
Terminated	Zeigt an, ob der Thread terminiert werden soll oder nicht.
ThreadID	ID des Threads.

Tabelle 28 Eigenschaften des Thread-Objektes

8.1.3 Methoden

Methode	Erklärung
Create	Erzeugt ein Thread-Objekt.
DoTerminate	Ruft den OnTerminate Eventhandler auf, beendet den Thread aber nicht.
Execute	Abstrakte Methode für den Code, den der Thread ausführt, wenn er gestartet wurde.
Resume	Versetzt den Thread in den zuteilungsfähigen Zustand.
Suspend	Versetzt den Thread in den angehaltenen Zustand.
Synchronize	Führt ein Methodeaufruf innerhalb des Haupt-VCL-Threads aus.
Terminate	Setzt die Eigenschaft Terminated auf True.

Tabelle 29 Methoden des Thread-Objektes

8.2 Beispiel einer Execute Methode

Hier geht es mir darum das Zusammenspiel der Eigenschaft Terminated und der Methode Terminate aufzuzeigen.

```

procedure TMyThreads.Execute;
begin
    // hier steht der Code, der vom Thread ausgeführt wird
    DoSomething();
end;

procedure TMyThreads.DoSomething;
var
    Loop: Integer;
begin
    for Loop := 0 to 1000000 do
        begin
            Inc(FCount);
            // ist der Flag Terminated gesetzt, Schleife verlassen (* Unit1)
            if Terminated then
                break;
            Synchronize(UpdateLVCaption);
        end;
    end;
end;

```

Die Methode `Terminate` setzt die Eigenschaft `Terminated` auf `True`. Damit sich der Thread nun beendet, ist es erforderlich, dass er in der `Execute` Methode periodisch die Eigenschaft `Terminated` prüft, und dann entsprechend darauf reagiert. Im Code Beispiel ist das die `if`-Abfrage in der `for`-Schleife der Methode `DoSomething`. Ist `Terminated` `True`, wird die Schleife verlassen, die `Execute`-Methode endet und damit auch der Thread.

8.3 Synchronisation des Thread-Objektes

Um einen Thread mit dem Haupt-VCL-Thread zu synchronisieren bietet die das Thread-Objekt die Methode `Synchronize` an. Nutzt man diese, erspart man sich das Einsetzen von kritischen Abschnitten, die natürlich trotzdem zur Verfügung stehen.

Der Methode `Synchronize` übergibt man eine Methode der Thread Klasse, in der eine Methode aus dem Haupt-VCL-Thread aufgerufen wird, welche ihrerseits nun Objekte im Haupt-VCL-Thread aktualisiert. Diese Konstruktion sieht nun wie folgt aus. Auch hier etwas Beispiel-Code aus dem beiliegenden Thread-Objekt Demo:

Aufruf von `Synchronize`:

```
procedure TMyThreads.DoSomething;
var
  Loop: Integer;
begin
  for Loop := 0 to 1000000 do
  begin
    [...]
    Synchronize(UpdateLVCaption);
  end;
end;
```

Und die zugehörige `Synchronize` Methode der Thread Klasse:

```
procedure TMyThreads.UpdateLVCaption;
begin
  Form1.UpdateLVCaption(FIndex, FCount);
end;
```

Und zu guter letzt die Methode `UpdateLVCaption` aus dem Haupt-VCL-Thread, welcher die Ausgabe im Listview aktualisiert:

```
procedure TForm1.UpdateLVCaption(Index, Count: Integer);
begin
  ListView1.Items[Index].SubItems[0] := 'Position: ' + IntToStr(Count);
  if Count = 10000 then
    ListView1.Items[Index].SubItems[0] := 'fertig';
end;
```

Wichtig ist, so lange die zugehörige `Synchronize` Methode aus dem Haupt-VCL-Thread ausgeführt wird, steht der Thread. Um nun zu verhindern, dass es zu Performance-Einbußen kommt, sollten die Synchronisations-Methoden so kurz wie möglich gehalten werden. Rufen

mehrere Threads gleichzeitig Synchronize auf werden die Aufrufe serialisiert, das heißt, sie werden vom Haupt-VCL-Thread nach einander ausgeführt, was wiederum bedeutet, dass alle Threads so lange warten, bis ihre Synchronize Methode aufgerufen und ausgeführt wurde.

→ Demo: TThreadObject